
Abstract

Theoretical results involving approximation of integrals are often established
from the construction and resultant manipulation of an appropriate kernel. The
systematic use of these kernels has produced an abundance of new approxi-
mations and error estimates in terms of norms of the integrand. Notwithstand-
ing the great success of this approach, many approximations and error results
have yet to be discovered due to the algebraic complexities involved; especially
those that involve product integrands.

We outline a method that uses the computer algebra system Maple that is
able to recapture the well known Ostrowksi, trapezoidal and Simpson’s inequal-
ities. Moreover, the technique, which involves manipulation of the Peano kernel,
can be adapted to develop new rules.

2000 Mathematics Subject Classification: Primary: 68W30, 65D30; Secondary:
26D10, 26D15.
Key words: Symbolic computation, Numerical integration, Simpson’s rule, Trape-

1. Introduction
Newton-Cotes and Newton-Cotes type integration has been an area of much
recent activity. Cerone, Dragomir, Pečaríc and others have reported on gener-
alizations of the Ostrowski, trapezoidal and Simpson rules. These results have
involved new bounds being expressed in different norms [1, 7, 9, 14, 16] as well
as new generalizations of the rules themselves [12, 15]. For a good overview
we refer the reader to the survey [1] and the book [11]. Much may be gar-
nished from the work of Cerone [2, 3, 4, 5, 6] whose results are a considerable
generalization of previous work.

Most of these results can be obtained via manipulation of the appropriate
Peano kernel and nearly all of the algebra required to establish these results
require partial integration, differentiation and some simplification - all of which
may be done by modern computer algebra systems (CAS’s) such as Maple [13],
Mathematica and MuPad. The algebraic techniques employed here are inspired
by the early work of Dragomir and Wang [8].

In this paper we begin by detailing the manual steps required to prove the
Ostrowksi inequality. In Sections3 and4 we show, with considerable detail,
how Maple can be used to prove the Ostrowksi and corrected trapezoidal in-
equalities. In Section5, Maple is used to establish a new corrected trape-
zoidal rule whose upper bound is 1.7 times smaller than the canonical cor-
rected trapezoidal rule. In Section6, Maple is used to recapture a recently
published modified Simpson rule [17]. Each of Sections3 – 6 are independent
Maple worksheets which the reader is encouraged to download fromhttp:
//www.roumeliotis.com.au/john/maple .

http://jipam.vu.edu.au/
mailto:John.Roumeliotis@vu.edu.au
http://jipam.vu.edu.au/
http://www.roumeliotis.com.au/john/maple

2. Ostrowski’s inequality – Manual Calculation
Ostrowski’s inequality, a generalized mid-point inequality, is quite simple to
establish. The proof involves partial integration of a piecewise linear Peano
kernel to obtain the rule and integration of the absolute value of the kernel to
obtain the bound.

The proof of Ostrowski’s inequality is well known and has been reproduced
many times in the literature. See for example the book [11] or the paper [10].

Theorem 2.1 (Ostrowski). Let f : I ⊆ R → R be a differentiable mapping
in Io (Io is the interior ofI), and leta, b ∈ Io with a < b. If f ′ : (a, b) →
R is bounded on(a, b), i.e., ‖f ′‖∞ := sup

t∈(a,b)

|f ′(t)| < ∞, then we have the

inequality:

(2.1)

∣∣∣∣∫ b

a

f(t) dt− (b− a)f(x)

∣∣∣∣ ≤
[

(b− a)2

4
+

(
x− a + b

2

)2
]
‖f ′‖∞

for all x ∈ (a, b).

Proof. Let the Peano kernelp be given by

p(t) =

{
t− a, a ≤ t ≤ x,

t− b, x < t ≤ b,

for somex ∈ [a, b]. Integrating the productp(t)f ′(t) gives

http://jipam.vu.edu.au/
mailto:John.Roumeliotis@vu.edu.au

I =

∫ b

a

p(t)f ′(t) dt(2.2)

=

∫ x

a

p(t)f ′(t) dt +

∫ b

x

p(t)f ′(t) dt(2.3)

= (x− a)f(x)−
∫ x

a

f(t) dt− (x− b)f(x)−
∫ b

x

f(t) dt(2.4)

= (b− a)f(x)−
∫ b

a

f(t) dt.(2.5)

Hölder’s inequality is used to obtain the bound.∣∣∣∣∫ b

a

f(t) dt− (b− a)f(x)

∣∣∣∣ = |I|(2.6)

=

∣∣∣∣∫ b

a

pf ′ dt

∣∣∣∣(2.7)

≤ ‖f ′‖∞
∫ b

a

|p| dt(2.8)

= ‖f ′‖∞
(∫ x

a

t− a dt +

∫ b

x

b− t dt

)
(2.9)

=

[
(b− a)2

4
+

(
x− a + b

2

)2
]
‖f ′‖∞.(2.10)

From a computer algebra perspective, the important steps are the integration by
parts in (2.3) to (2.4) and integration of the positive kernel in (2.8) to (2.10).

http://jipam.vu.edu.au/

For complex kernels, it is often difficult to ascertain whether a kernel is positive
(as in equation (2.9)), and certainly cannot be automatically determined by a
CAS.

In the next section we show the Maple steps used to establish (2.1). It is
helpful to compare each step with those in (2.2)–(2.10) above.

http://jipam.vu.edu.au/
mailto:John.Roumeliotis@vu.edu.au

3. Ostrowski Inequality
In this section, we will use the Maple computer algebra system to derive the well
known Ostrowski inequality. This section is an actual Maple worksheet called
ostrowski.mwswhich you can download fromhttp://www.roumeliotis.
com.au/john/maple .

> restart;

> with( student ):

The Ostrowski inequality is a first order approximation for a finite integral
using a general interior pointx. Hence the kernel is of first order and vanishes
at the pointst = a andt = b. Define the general kernelp(t, x) = t− x where
a ≤ x andx ≤ b

> p := (t,x) -> t-x;

p := (t, x) → t− x

Integrating the productp(t) ( d
dt

f(t)) by parts and simplifying produces the
approximation.

> Ostr_rule := intparts( Int(
> p(t,a)*D(f)(t) , t=a..x ) , p(t,a) ) +
> intparts( Int( p(t,b) * D(f)(t) , t=x..b ) ,
> p(t,b) );

Ostr_rule := (x− a) f(x)−
∫ x

a

f(t) dt− (x− b) f(x)−
∫ b

x

f(t) dt

> Ostr_rule := collect( combine( Ostr_rule ) , f );

Ostr_rule := (b− a) f(x) +

∫ b

a

− f(t) dt

http://jipam.vu.edu.au/
mailto:John.Roumeliotis@vu.edu.au
http://jipam.vu.edu.au/
http://www.roumeliotis.com.au/john/maple
http://www.roumeliotis.com.au/john/maple

To obtain the bound, via Hölder’s inequality, we need to integrate|p(t)|.
> Ostr_bound := int( p(t,a) , t=a..x )
> - int( p(t,b) , t=x..b );

Ostr_bound := x2 − a2

2
− a (x− a)− b2

2
+ b (−x + b)

Completing the square and factorizing the constant term produces the result
> Ostr_bound := op( 1, completesquare
> (Ostr_bound , x )) + factor ( Ostr_bound -
> op ( 1, completesquare ( Ostr_bound , x )) );

Ostr_bound := (x− a

2
− b

2
)2 +

(a− b)2

2

http://jipam.vu.edu.au/
mailto:John.Roumeliotis@vu.edu.au

4. Corrected Trapezoidal Rule
In this section, we will use the Maple computer algebra system to derive the
corrected trapezoidal rule using a Peano kernel approach. This section is an
actual Maple worksheet calledcorrected_trap.mwswhich you can download
from http://www.roumeliotis.com.au/john/maple .

The corrected rule is fourth order and is defined over one interval[a, b], this
contrasts with the Ostrowski kernel which has separate definitions for [a,x] and
(x,b]. Recall that the corrected trapezoidal rule is

(4.1)
∫ b

a

f(x) dx =
(b− a) f(a)

2
+

(b− a) f(b)

2

+
(b− a)2 D(f)(a)

12
− (b− a)2 D(f)(b)

12
+ E

where

|E| ≤
max(

∣∣(D(4))(f)
∣∣)(b− a)5

720
.

Define an arbitrary fourth order polynomial as our Peano kernel and solve
for the coefficients of the kernel by specifying the required coefficients of the
rule. To begin,restart the Maple kernel and load thestudent package
which is required for the following calculations.

> restart; with( student ):

http://jipam.vu.edu.au/
mailto:John.Roumeliotis@vu.edu.au
http://jipam.vu.edu.au/

Integral Inequalities and
Computer Algebra Systems

John Roumeliotis

Title Page

Contents

JJ II

J I

Go Back

Close

Quit

Page 11 of 44

J. Ineq. Pure and Appl. Math. 6(5) Art. 141, 2005

http://jipam.vu.edu.au

Define the general fourth order Peano kernel,p(t).

> p := t -> 1/4!*(t-A[1])*(t-A[2])
> *(t-A[3])*(t-A[4]);

p := t → (t− A1) (t− A2) (t− A3) (t− A4)

4!

DefineI _rule to be the integral of the product(D(4))(f)(t) p(t).

> I_rule := Int( (D@@4)(f)(t) * p(t) , t=a..b );

I _rule :=

∫ b

a

1

24
(D(4))(f)(t) (t− A1) (t− A2) (t− A3) (t− A4) dt

This is integrated by parts four times. The Maple commandapplyop is
used to integrate the required part of the expression.

> intparts( I_rule , p(t) ):
> applyop( intparts , nops(%) ,
> % , diff( p(t),t ) ):
> applyop( intparts , nops(%) ,
> % , diff( p(t),t$2 ) ):
> Corr_Rule := applyop( intparts , nops(%) ,
> % , diff( p(t),t$3 ));

http://jipam.vu.edu.au/
mailto:John.Roumeliotis@vu.edu.au

Corr_Rule :=
1

24
(b− A1) (b− A2) (b− A3) (b− A4) (D(3))(f)(b)

− 1

24
(a− A1) (a− A2) (a− A3) (a− A4) (D(3))(f)(a)− (

1

24
(b− A2) (b− A3) (b− A4) +

1

24
(b− A1) (b− A3) (b− A4)

+
1

24
(b− A1) (b− A2) (b− A4) +

1

24
(b− A1) (b− A2) (b− A3))(D

(2))(f)(b)

+ (
1

24
(a− A2) (a− A3) (a− A4) +

1

24
(a− A1) (a− A3) (a− A4)

+
1

24
(a− A1) (a− A2) (a− A4) +

1

24
(a− A1) (a− A2) (a− A3))(D

(2))(f)(a)

+ (
1

12
(b− A3) (b− A4) +

1

12
(b− A2) (b− A4) +

1

12
(b− A2) (b− A3)

+
1

12
(b− A1) (b− A4) +

1

12
(b− A1) (b− A3) +

1

12
(b− A1) (b− A2))D(f)(b)

− (
1

12
(a− A3) (a− A4) +

1

12
(a− A2) (a− A4) +

1

12
(a− A2) (a− A3)

+
1

12
(a− A1) (a− A4) +

1

12
(a− A1) (a− A3)

+
1

12
(a− A1) (a− A2))D(f)(a)− (b− 1

4
A4 −

1

4
A3 −

1

4
A2 −

1

4
A1) f(b)

+(a− 1

4
A4 −

1

4
A3 −

1

4
A2 −

1

4
A1) f(a) +

∫ b

a

f(t) dt

To re-capture the corrected trapezoidal integration rule, we solve for

http://jipam.vu.edu.au/
mailto:John.Roumeliotis@vu.edu.au

A1, A2, A3, A4

such that the coefficients ofd
2

dt2
f(t) and d3

dt3
f(t) vanish att = a andt = b.

> eq1 := coeff( Corr_Rule , (D@@2)(f)(a) ) = 0;

eq1 :=
1

24
(a− A2) (a− A3) (a− A4) +

1

24
(a− A1) (a− A3) (a− A4)

+
1

24
(a− A1) (a− A2) (a− A4) +

1

24
(a− A1) (a− A2) (a− A3) = 0

> eq2 := coeff( Corr_Rule , (D@@2)(f)(b) ) = 0;

eq2 := − 1

24
(b− A2) (b− A3) (b− A4)−

1

24
(b− A1) (b− A3) (b− A4)

− 1

24
(b− A1) (b− A2) (b− A4)−

1

24
(b− A1) (b− A2) (b− A3) = 0

> eq3 := coeff( Corr_Rule , (D@@3)(f)(a) ) = 0;

eq3 := − 1

24
(a− A1) (a− A2) (a− A3) (a− A4) = 0

> eq4 := coeff( Corr_Rule , (D@@3)(f)(b) ) = 0;

eq4 :=
1

24
(b− A1) (b− A2) (b− A3) (b− A4) = 0

> solve( {eq1,eq2,eq3,eq4} , {A[1],A[2],A[3],A[4]}
);

{A4 = a, A3 = b, A1 = b, A2 = a}, {A4 = a, A3 = b, A2 = b, A1 = a},
{A4 = a, A1 = b, A2 = b, A3 = a}, {A1 = b, A2 = a, A3 = a, A4 = b},
{A2 = b, A1 = a, A3 = a, A4 = b}, {A3 = b, A2 = a, A1 = a, A4 = b}

http://jipam.vu.edu.au/

Without loss of generality we can choose the first solution and weassign
the variablesA1, A2, A3, A4 to this.

> assign( %[1] );

The integration rule is
> Corr_Rule;

1

12
(b− a)2 D(f)(b)− 1

12
(−b + a)2 D(f)(a)

−(
b

2
− a

2
) f(b) + (

a

2
− b

2
) f(a) +

∫ b

a

f(t) dt

To obtain the error bound, we employ a Hölder inequality∫ b

a

∣∣p(t) (D(4))(f)(t)
∣∣ dt ≤ max(

∣∣(D(4))(f)
∣∣) ∫ b

a

|p(t)| dt

which requires calculation of|p| . It is easy to see that0 < p

> p(t);

(t− b)2 (t− a)2

24
so calculation of the bound is a simple matter of integration
> int( p(t) , t=a..b );

b5

120
− a5

120
+

(− b

12
− a

12
) (b4 − a4)

4
+

(
1

24
b2 +

1

6
b a +

1

24
a2) (b3 − a3)

3

+
(− 1

12
b2 a− 1

12
a2 b) (b2 − a2)

2
+

b2 a2 (b− a)

24

http://jipam.vu.edu.au/

> factor( % );

−(−b + a)5

720

http://jipam.vu.edu.au/
mailto:John.Roumeliotis@vu.edu.au

5. New Corrected Trapezoidal Rule
In this section, we will use the Maple computer algebra system to derive a new
corrected trapezoidal rule using a Peano kernel approach. This section is an ac-
tual Maple worksheet callednew_corrected_trap.mwswhich you can down-
load fromhttp://www.roumeliotis.com.au/john/maple .

The corrected trapezoidal rule in Section 4 was obtained by determining the
coefficients of the Peano kernel so that thed2

dt2
f(t) and d3

dt3
f(t) terms vanish in

the integration rule. Thed
dt

f(t) terms are equal in magnitude and opposite in
sign, so that they cancel everywhere except at the boundary in a composite rule.
There is an opportunity to create a further correction to the corrected trapezoidal
rule by imposing the condition that all three derivatives cancel for a composite
rule. This would leave one free parameter which can be used to minimize the
error bound. Care must be taken when integrating the positive kernel to produce
the error bound, since determining the sign of the kernel cannot be automated.

> restart;

> with(student):

Call in special plotting packages for 3D plots to be produced in this section.

> with( plots ): with( plottools ):

Warning, the name changecoords has been redefined

Warning, the assigned name arrow now has a global binding

http://jipam.vu.edu.au/
mailto:John.Roumeliotis@vu.edu.au

As with the corrected rule, the kernel is a fourth order polynomial since this
rule will have fourth order accuracy.

> p := t->1/4!*(t-A[1])*(t-A[2])
> *(t-A[3])*(t-A[4]);

p := t → (t− A1) (t− A2) (t− A3) (t− A4)

4!

DefineCorr_Rule to be the integral of the product(D(4))(f)(t) p(t).

> Corr_Rule := Int( (D@@4)(f)(t) * p(t) ,
> t=a..b );

Corr_Rule :=

∫ b

a

1

24
(D(4))(f)(t) (t− A1) (t− A2) (t− A3) (t− A4) dt

Integrate by parts four times.

> intparts( Corr_Rule , p(t) ):
> applyop( intparts , nops(%) ,
> % , diff( p(t),t ) ):
> applyop( intparts , nops(%) ,
> % , diff( p(t),t$2 ) ):
> Corr_Rule := applyop( intparts , nops(%) ,
> % , diff( p(t),t$3 ));

Corr_Rule :=
1

24
(b− A1) (b− A2) (b− A3) (b− A4) (D(3))(f)(b)

− 1

24
(a− A1) (a− A2) (a− A3) (a− A4) (D(3))(f)(a)− (

1

24
(b− A2) (b− A3) (b− A4) +

1

24
(b− A1) (b− A3) (b− A4)

+
1

24
(b− A1) (b− A2) (b− A4) +

1

24
(b− A1) (b− A2) (b− A3))(D

(2))(f)(b)

+ (
1

24
(a− A2) (a− A3) (a− A4) +

1

24
(a− A1) (a− A3) (a− A4)

+
1

24
(a− A1) (a− A2) (a− A4) +

1

24
(a− A1) (a− A2) (a− A3))(D

(2))(f)(a)

+ (
1

12
(b− A3) (b− A4) +

1

12
(b− A2) (b− A4) +

1

12
(b− A2) (b− A3)

+
1

12
(b− A1) (b− A4) +

1

12
(b− A1) (b− A3) +

1

12
(b− A1) (b− A2))D(f)(b)

− (
1

12
(a− A3) (a− A4) +

1

12
(a− A2) (a− A4) +

1

12
(a− A2) (a− A3)

+
1

12
(a− A1) (a− A4) +

1

12
(a− A1) (a− A3)

+
1

12
(a− A1) (a− A2))D(f)(a)− (b− 1

4
A4 −

1

4
A3 −

1

4
A2 −

1

4
A1) f(b)

+(a− 1

4
A4 −

1

4
A3 −

1

4
A2 −

1

4
A1) f(a) +

∫ b

a

f(t) dt

Impose the condition that the sum of the coefficient of each derivative at

t = a andt = b vanishes.
> eq1 := coeff( Corr_Rule , D(f)(a) )
> = -coeff( Corr_Rule , D(f)(b) );
> eq2 := coeff( Corr_Rule , (D@@2)(f)(a) )
> = -coeff( Corr_Rule ,(D@@2)(f)(b) );
> eq3 := coeff( Corr_Rule , (D@@3)(f)(a) )
> = -coeff( Corr_Rule ,(D@@3)(f)(b) );

eq1 := − 1

12
(a− A3) (a− A4)−

1

12
(a− A2) (a− A4)−

1

12
(a− A2) (a− A3)

− 1

12
(a− A1) (a− A4)−

1

12
(a− A1) (a− A3)−

1

12
(a− A1) (a− A2)

= − 1

12
(b− A3) (b− A4)−

1

12
(b− A2) (b− A4)−

1

12
(b− A2) (b− A3)

− 1

12
(b− A1) (b− A4)−

1

12
(b− A1) (b− A3)−

1

12
(b− A1) (b− A2)

eq2 :=
1

24
(a− A2) (a− A3) (a− A4) +

1

24
(a− A1) (a− A3) (a− A4)

+
1

24
(a− A1) (a− A2) (a− A4) +

1

24
(a− A1) (a− A2) (a− A3)

=
1

24
(b− A2) (b− A3) (b− A4) +

1

24
(b− A1) (b− A3) (b− A4)

+
1

24
(b− A1) (b− A2) (b− A4) +

1

24
(b− A1) (b− A2) (b− A3)

eq3 := − 1

24
(a− A1) (a− A2) (a− A3) (a− A4)

= − 1

24
(b− A1) (b− A2) (b− A3) (b− A4)

Solve the non-linear system to find the first three terms,A1, A2, A3 of p(t).
> solve( {’eq||i’$i=1..3} , {A[i]$i=1..3} );

{A3 = %1, A1 = a−%1 + b, A2 = a + b− A4},
{A3 = a + b− A4, A1 = a−%1 + b, A2 = %1},
{A3 = %1, A2 = a−%1 + b, A1 = a + b− A4}
%1 := RootOf(_Z 2 + (−a− b) _Z − a A4 + A4

2 − b A4 + 2 b a)
Without loss of generality we can choose the first solution. The three solu-

tions here appear since we have not imposed any ordering betweenA1, A2 and
A3.

> Sols := allvalues( %[1] );

Sols := {A3 =
a

2
+

b

2
+

1

2
%1, A1 =

a

2
+

b

2
− 1

2
%1, A2 = a + b− A4},

{A3 =
a

2
+

b

2
− 1

2
%1, A2 = a + b− A4, A1 =

a

2
+

b

2
+

1

2
%1}

%1 :=
√

a2 − 6 b a + b2 + 4 a A4 − 4 A4
2 + 4 b A4

> assign( %[1] );

Thus the new corrected trapezoidal rule appears below. Note that the sec-
ond derivative has vanished. This is typical of rules of this type and it can be
shown that all even derivatives vanish in "corrected" rules. The Maple command

collect is used to collect all thef terms together, i.e.f(a), f(b), D(f)(a), D(f)(b),
etc.map(factor, applies thefactor function across each of these terms.

> map(factor,collect(Corr_Rule , f ));

1

2
(a− b) f(b) +

1

2
(a− b) f(a) +

1

12
(a− b)2 D(f)(b)

+
1

96
(−b + a−%1) (a− A4) (−b + a + %1) (−b + A4) (D(3))(f)(b)

+
1

96
(−b + a + %1) (b− A4) (−b + a−%1) (a− A4) (D(3))(f)(a)

− 1

12
(a− b)2 D(f)(a) +

∫ b

a

f(t) dt

%1 :=
√

a2 − 6 b a + b2 + 4 a A4 − 4 A4
2 + 4 b A4

The kernel is.
> p(t);

1

24
(t− a

2
− b

2
+

1

2

√
a2 − 6 b a + b2 + 4 a A4 − 4 A4

2 + 4 b A4) (t− a− b + A4)

(t− a

2
− b

2
− 1

2

√
a2 − 6 b a + b2 + 4 a A4 − 4 A4

2 + 4 b A4) (t− A4)

The term under the radical imposes a restricted domain onA4. We can select
this term and complete the square to determine this domain.

Select the term.
> op([2,4,2,1],p(t));

a2 − 6 b a + b2 + 4 a A4 − 4 A4
2 + 4 b A4

Complete the square.
> completesquare( % , A[4] );

−4 (A4 −
a

2
− b

2
)2 + 2 a2 − 4 b a + 2 b2

Notice that the remaining term is a perfect square, so we factor this.
> factor(subsop(1=0,%));

2 (a− b)2

> fac := op(1,%%)+%;

fac := −4 (A4 −
a

2
− b

2
)2 + 2 (a− b)2

It is obvious, by inspection, that the above expression is non-negative on a
domain larger than[a, b]. We can find this domain.

> solve( fac=0 , A[4] );
1

2
a +

1

2
b− 1

2

√
2 a +

1

2

√
2 b,

1

2
a +

1

2
b +

1

2

√
2 a− 1

2

√
2 b

> A4sol := map( factor , [ % ] );

A4sol := [−(
√

2− 1) (a− 3 b− 2
√

2 b)

2
,

(1 +
√

2) (a− 3 b + 2
√

2 b)

2
]

> A4min := A4sol[2]; A4max := A4sol[1];

A4min :=
(1 +

√
2) (a− 3 b + 2

√
2 b)

2

A4max := −(
√

2− 1) (a− 3 b− 2
√

2 b)

2

The domain ofA4 is [A4min, A4max ] whereA4min < a andb < A4max .
We can now inspect the nature of the kernel over the unit interval.

> A4min01 := expand(subs( a=0,b=1, A4min ));

A4min01 :=
1

2
−
√

2

2
> A4max01 := expand(subs( a=0,b=1, A4max ));

A4max01 :=
1

2
+

√
2

2
The following animated and 3D plots reveal thatp generally changes sign

twice asA4 increases fromA4minto A4max. We already know the roots of
p, they are the termsA1, A2, A3 andA4 and so to determine the sign of the
kernel for anyt we require knowledge of how theAi, i = 1..4 relate to each
other and the interval[a, b].

> animate( plot , [ subs(a=0,b=1,p(t)) , t=0..1 ],
> A[4]=A4min01..A4max01 , frames=60 );

A[4] = -.20711

0.002

0.001

0

0.001

0.002

0.2 0.4 0.6 0.8 1
t

3D plot of the kernel.
> plot1 := contourplot( subs(a = 0, b = 1, p(t) ),
t=0..1, A[4] = A4min01..A4max01 , contours = 15 ):
> plot2 := plot3d ( subs ( a=0, b=1, p(t) ),
> t = 0 .. 1, A[4] = A4min01 .. A4max01,
> grid = [ 30, 30 ], axes = boxed ) :

> ptrans := transform((x,y) -> [x,y,-0.006]):
> display ( { ptrans ( plot1) , plot2} ,
> orientation = [ -135 , 65 ] ) ;

0
0.2

0.4
0.6

0.8
1

00.20.40.60.811.2
0.006

0.004

0.002

0

0.002

The following plot reveals that the kernel coefficients change ordering across
four regions. They are:

A[4]<a<A[1] <A[3]<b< A[2], for A4min<A[4]<a

A[1]<a<A[4] <A[2]<b< A[3], for a<A[4]<(a+b)/2

A[1]<a<A[2] <A[4]<b< A[3], for (a+b)/2<A[4]<b

A[2]<a<A[1] <A[3]<b< A[4], for b<A[4]<A4max
> plot( subs(a=0,b=1,[A[i]$i=1..4]) ,
A[4]=A4min01..A4max01 ,
thickness=4,color=[blue,black,yellow,red]);

–0.2

0

0.2

0.4

0.6

0.8

1

1.2

–0.2 0.2 0.4 0.6 0.8 1 1.2
A[4]

A more descriptive plot than the Maple graphic above is shown in Figure1.
The upper boundIb, which is the integral of the positive kernel, requires four

separate calculations according to the regions defined above.

A[4]<a<A[1] <A[3]<b< A[2], for A4min<A[4]<a
> Ib1 := factor( -int( p(t) , t=a..A[1] )
+ int( p(t) , t=A[1]..A[3] )
- int( p(t) , t=A[3]..b ) ):

A[1]<a<A[4] <A[2]<b< A[3], for a<A[4]<(a+b)/2
> Ib2 := factor( -int( p(t) , t=a..A[4] )
+ int( p(t) , t=A[4]..A[2] )
- int( p(t) , t=A[2]..b ) ):

A[1]<a<A[2] <A[4]<b< A[3], for (a+b)/2<A[4]<b
> Ib3 := factor( -int( p(t) , t=a..A[2] )
+ int( p(t) , t=A[2]..A[4] )
- int( p(t) , t=A[4]..b ) ):

A[2]<a<A[1] <A[3]<b< A[4], for b<A[4]<A4max
> Ib4 := Ib1:

Hence, we can defineIb in a piecewise fashion.
> Ib := piecewise( A[4]<a , Ib1 , A[4]<(a+b)/2 ,
Ib2 , A[4] < b , Ib3 , A[4]<A4max , Ib4 , 0 ):

As a function ofA4, the upper bound reveals four minimum points. We will
explore the first minimum point and the new corrected rule resulting from eval-
uatingA4 at this point. The results are the same for all four minima.

> plot( subs(a=0,b=1,Ib) , A[4]=A4min01..A4max01 );

0.0009

0.001

0.0011

0.0012

0.0013

0.0014

–0.2 0 0.2 0.4 0.6 0.8 1 1.2
A[4]

The first minimum point is a stationary point ofIb1 .
> assume( b>a );

> solve(diff( Ib1,A[4]) , A[4]);

b˜,
1

2
b˜ +

1

2
a˜− 1

4

√
7 a˜ +

1

4

√
7 b˜,

1

2
b˜ +

1

2
a˜ +

1

4

√
7 a˜− 1

4

√
7 b˜,

a˜

2
+

b˜

2
, a˜

Select the correct point since forIb1 we must haveA4min < A4 andA4 < a.
> sol1 :=[ % ][3];

sol1 :=
1

2
b˜ +

1

2
a˜ +

1

4

√
7 a˜− 1

4

√
7 b˜

Hence the upper bound is

> factor( subs(A[4]=sol1 , Ib1 ) );

−5 (a˜− b˜)5

6144
and the new corrected trapezoidal rule is

> map( factor@expand , subs(A[4]=sol1,Corr_Rule) );

− 3

2048
(D(3))(f)(b˜) (a˜− b˜)4 +

3

2048
(D(3))(f)(a˜) (a˜− b˜)4

+
1

12
D(f)(b˜) (a˜− b˜)2 − 1

12
D(f)(a˜) (a˜− b˜)2

+
1

2
f(b˜) (a˜− b˜) +

1

2
f(a˜) (a˜− b˜) +

∫ b˜

a˜

f(t) dt

When compared to the generic corrected trapezoidal rule∫ b

a

f(x) dx =
(b− a)f(a)

2
+

(b− a)f(b)

2

+
(b− a)2 D(f)(a)

12
− (b− a)2D(f)(b)

12
+ E,

where

|E| ≤
max(

∣∣(D(4))(f)
∣∣) (b− a)5

720

we can see that the new rule has two added third derivative function evaluations
(which will cancel on composition) and an upper bound coefficient of 5/6144.
This is approximately 1.7 times better.

> evalf( (1/720) / (5/6144) );

1.706666667

In practice, if we had knowledge of the 3rd derivative on the boundary then we
would use the 6th order corrected trapezoidal rule

∣∣∣∣−(a− b)4 (D(3))(f)(b)

720
+

(a− b)4 (D(3))(f)(a)

720
+

(a− b)2D(f)(b)

12

− (a− b)2D(f)(a)

12
+

(a− b) f(b)

2
+

(a− b)f(a)

2
+

∫ b

a

f(t) dt

∣∣∣∣
≤

max(
∣∣(D(6))(f)

∣∣) (b− a)7

30240
.

Nevertheless, we have shown that Maple is a useful tool for exploring quadra-
ture rules and their error bounds.

A4

A4

A2

A1

A3

A4,min

A4,min

A4,max

A4,maxa

a

b

b

R1 R2 R3 R4

Figure 1: A graphic showing the ordering ofA1, A2, A3, A4 in regions
R1, R2, R3, R4.

http://jipam.vu.edu.au/
mailto:John.Roumeliotis@vu.edu.au
http://jipam.vu.edu.au/


Integral Inequalities and
Computer Algebra Systems

John Roumeliotis

Title Page

Contents

JJ II

J I

Go Back

Close

Quit

Page 31 of 44

J. Ineq. Pure and Appl. Math. 6(5) Art. 141, 2005

http://jipam.vu.edu.au

6. Modified Simpson’s Rule
In this section, we will use the Maple computer algebra system to derive a
modified Simpson rule that was recently established. This section is an ac-
tual Maple worksheet calledmod_simp_6.mwswhich you can download from
http://www.roumeliotis.com.au/john/maple .

Ujević and Roberts (2004) recently established the following modified Simp-
son type rule:∫ b

a

f(x)dx =
(b− a) (7f(a) + 16f(a+b

2
) + 7f(b))

30

− (b− a)2 (D(f)(b)−D(f)(a))

60
+ E,

where|E| ≤ max(|(D(6))(f)|) (b−a)7

604800
. We can see that the weights are different to

those of the well known Simpson inequality and that the end-point derivative
correction produces a rule of order 6.

Using Maple, we will recapture this result by employing a split 6th order
Peano kernel. In order to eliminate the presence of higher derivatives at the
end-points, the kernel will require the correct smoothness att=a,b.

> restart;

> with( student ):

The (t− a)4 and(t− b)4 terms will ensure no derivatives higher than one
exist at the endpoints. Since this will be a sixth order rule, the kernel needs to
be a sixth order polynomial. Thus we have four free parametersc0, c1, c2, c3.

> p1 := 1/6!*(t-a)^4*(t^2+c0*t+c1);

> p2 := 1/6!*(t-b)^4*(t^2+c2*t+c3);

p1 :=
(t− a)4 (t2 + c0 t + c1 )

720

p2 :=
(t− b)4 (t2 + c2 t + c3 )

720
> I1 := Int( p1 * (D@@6)(f)(t) , t=a..(a+b)/2 );

> I2 := Int( p2 * (D@@6)(f)(t) , t=(a+b)/2..b );

I1 :=

∫ a

2
+

b

2

a

1

720
(t− a)4 (t2 + c0 t + c1 ) (D(6))(f)(t) dt

I2 :=

∫ b

a

2
+

b

2

1

720
(t− b)4 (t2 + c2 t + c3 ) (D(6))(f)(t) dt

Integrate by parts six times.
> I1 := intparts( I1 , p1 ):

> I2 := intparts( I2 , p2 ):
> for i from 1 to 5 do
> I1 := applyop( intparts , nops(I1) ,
> I1 , diff(p1,t$i) ):
> I2 := applyop( intparts , nops(I2) ,
> I2 , diff(p2,t$i) )
> end do:

We can see that the integration rule contains sampling up to order one at the
endpoints and sampling up to order 5 at the midpoint. The midpoint sampling

is due to the fact that the Peano kernel is discontinuous here.
> I_MS := map( factor , collect( combine( I1+I2 ) ,
f ) );

I _MS :=
1

23040
(a− b)4 (−c2 a + c0 a− 2 c3 + 2 c1

+c0 b− c2 b) (D(5))(f)(
a

2
+

b

2
) +

1

11520
(a− b)3(4 a2 + 3 c0 a + 5 c2 a

+8 a b + 8 c1 + 8 c3 + 5 c0 b + 4 b2 + 3 c2 b)(D(4))(f)(
a

2
+

b

2
)

− 1

1440
(a− b)2 (4 a2 − c0 a + 5 c2 a− 6 c1 − 5 c0 b + 6 c3 + c2 b

−4 b2) (D(3))(f)(
a

2
+

b

2
) +

1

240
(a− b) (4 a2 − c0 a

+5 c2 a + 5 c0 b + 4 c3 − c2 b + 4 b2 + 4 c1 )(D(2))(f)(
a

2
+

b

2
)

− 1

60
(5 c2 a− 3 c2 b + 2 c3 + 8 a2 − 8 b2 + 3 c0 a− 5 c0 b

−2 c1 ) D(f)(
a

2
+

b

2
)− 1

30
(a2 + c0 a + c1 ) D(f)(a)

+
1

30
(b2 + c2 b + c3 ) D(f)(b) +

1

6
(4 a− 4 b− c0 + c2 ) f(

a

2
+

b

2
)

+

∫ b

a

f(t) dt− 1

6
(2 b + c2 ) f(b) +

1

6
(2 a + c0 ) f(a)

Choosec0, c1, c2, c3so that all sampling of derviatives at the midpoint
vanish.

http://jipam.vu.edu.au/
mailto:John.Roumeliotis@vu.edu.au
http://jipam.vu.edu.au/


Integral Inequalities and
Computer Algebra Systems

John Roumeliotis

Title Page

Contents

JJ II

J I

Go Back

Close

Quit

Page 34 of 44

J. Ineq. Pure and Appl. Math. 6(5) Art. 141, 2005

http://jipam.vu.edu.au

> eq1 := map( factor , collect( coeff( I_MS ,
> D(f)((a+b)/2) ) , [c0,c1,c2,c3] ) ) = 0;

eq1 := −(3 a− 5 b) c0

60
+

c1

30
− (5 a− 3 b) c2

60
+

2 b2

15
− c3

30
− 2 a2

15
= 0

> eq2 := collect( coeff( I_MS ,
> (D@@2)(f)((a+b)/2) ) , [c0,c1,c2,c3] ) = 0;

eq2 :=
(a− b) (−a + 5 b) c0

240
+ (

a

60
− b

60
) c1 +

(a− b) (5 a− b) c2

240

+(
a

60
− b

60
) c3 +

(a− b) (4 a2 + 4 b2)

240
= 0

> eq3 := collect( coeff( I_MS ,
> (D@@3)(f)((a+b)/2) ) , [c0,c1,c2,c3] ) = 0;

eq3 := −(a− b)2 (−a− 5 b) c0

1440
+

(a− b)2 c1

240
− (a− b)2 (5 a + b) c2

1440

−(a− b)2 c3

240
− (a− b)2 (4 a2 − 4 b2)

1440
= 0

> eq4 := collect( coeff( I_MS ,
> (D@@4)(f)((a+b)/2) ) , [c0,c1,c2,c3] ) = 0;

eq4 :=
(a− b)3 (3 a + 5 b) c0

11520
+

(a− b)3 c1

1440
+

(a− b)3 (5 a + 3 b) c2

11520

+
(a− b)3 c3

1440
+

(a− b)3 (4 a2 + 8 a b + 4 b2)

11520
= 0

> eq5 := collect( coeff( I_MS ,
> (D@@5)(f)((a+b)/2) ) , [c0,c1,c2,c3] ) = 0;

eq5 :=
(a− b)4 (a + b) c0

23040
+

(a− b)4 c1

11520

+
(a− b)4 (−a− b) c2

23040
− (a− b)4 c3

11520
= 0

Solve to obtain the coefficients
> solve( {eq1,eq2,eq3,eq4} , {c0,c1,c2,c3} );

{c2 = −7 a

5
− 3 b

5
, c3 =

1

2
a2 +

2

5
a b +

1

10
b2,

c0 = −3 a

5
− 7 b

5
, c1 =

1

10
a2 +

2

5
a b +

1

2
b2}

> assign( % );

Hence, the integration rule is:
> I_MS := map( factor , I_MS );

I _MS := − 1

60
(a− b)2 D(f)(a) +

1

60
(a− b)2 D(f)(b)

+
8

15
(a− b) f(

a

2
+

b

2
) +

∫ b

a

f(t) dt

+
7

30
(a− b) f(b) +

7

30
(a− b) f(a)

and the split kernel is
> p1 := factor( p1 );

p1 :=
(−t + a)4 (10 t2 − 6 t a− 14 t b + a2 + 4 a b + 5 b2)

7200

> p2 := factor( p2 );

p2 :=
(t− b)4 (10 t2 − 14 t a− 6 t b + 5 a2 + 4 a b + b2)

7200
Completing the square in bothp1andp2shows that the kernel is positive
> p1 := applyop( completesquare , 3 , p1 , t );

p1 :=
(−t + a)4 (10 (t− 3 a

10
− 7 b

10
)2 +

a2

10
− a b

5
+

b2

10
)

7200
> p1 := subsop( 3 = op([3,1],p1) + factor(op(3,p1)
> - op([3,1],p1)) , p1 );

p1 :=
(−t + a)4 (10 (t− 3 a

10
− 7 b

10
)2 +

(a− b)2

10
)

7200
> p2 := applyop( completesquare , 3 , p2 , t );

p2 :=
(t− b)4 (10 (t− 7 a

10
− 3 b

10
)2 +

a2

10
− a b

5
+

b2

10
)

7200
> p2 := subsop( 3 = op([3,1],p2) + factor(op(3,p2)
> - op([3,1],p2)) , p2 );

p2 :=
(t− b)4 (10 (t− 7 a

10
− 3 b

10
)2 +

(a− b)2

10
)

7200
Thus the upper bound is calculated directly from the integral of the kernel
> factor( int( p1 , t=a..(a+b)/2 )
> + int( p2 , t=(a+b)/2..b ) );

−(a− b)7

604800
We have used this Maple session to show how one may apply a computer

algebra system such as Maple to easily recpature known results and extend to
new integration rules.

http://jipam.vu.edu.au/
mailto:John.Roumeliotis@vu.edu.au
http://jipam.vu.edu.au/


Integral Inequalities and
Computer Algebra Systems

John Roumeliotis

Title Page

Contents

JJ II

J I

Go Back

Close

Quit

Page 38 of 44

J. Ineq. Pure and Appl. Math. 6(5) Art. 141, 2005

http://jipam.vu.edu.au

7. Other Results
In this section we briefly state other results obtained using the technique out-
lined in Sections2 – 6. For each result, we state the general Peano kernel
complete with free parameters and an outline of the steps taken to produce the
integral inequality.

1. Fourth Order Modified Simpson Inequality. The kernel below will pro-
duce a fourth order rule with sampling of all derivatives at the discontinu-
ous pointx and only sampling off at the endpointst = a, b.

p(t) =


1

4!
(t− a)3(t− a0), t ∈ [a, x),

1

6!
(t− b)3(t− b0), t ∈ [x, b].

Given this kernel the motivation is to produce an optimal rule. That is,
minimize the upper bound using similar steps as shown in Section5. The
result is∣∣∣∣∫ b

a

f(t) dt− b− a

16

(
23/4f(a) + 2

(
2− 21/4

) (√
(2) + 2 21/4 + 4

)
×f

(
a + b

2

)
+

(b− a)2

12

(
8− 3 23/4

)
f ′′

(
a + b

2

)
+ 23/4f(b)

)∣∣∣∣
≤ 2− 23/4

3840
(b− a)5‖f (4)‖∞

2. Sixth Order Modified Simpson Inequality. Multiply each branch of the
well known Simpson kernel with an arbitrary quadratic polynomial.

p(t) =


1

6!
(t− a)3(t− a+2b

3
)(t2 + a1t + a0), t ∈ [a, a+b

2
)

1

6!
(t− b)3(t− 2a+b

3
)(t2 + b1t + b0), t ∈ [a+b

2
, b]

Find the coefficients so that rule is corrected, that is end-point derivatives
are equal and opposite, and as many midpoint evaluations vanish as possi-
ble. The result is

(7.1)

∣∣∣∣∣
∫ b

a

f(t) dt−

[
b− a

18

(
5f(a) + 8f

(
a + b

2

)
+ 5f(b)

)

+
(b− a)2

45

(
f ′(a)− f ′(b)

)
− (b− a)3

180
f ′′

(
a + b

2

)]∣∣∣∣∣
≤ 19

7257600
(b− a)7‖f (6)‖∞

3. Eighth Order Modified Simpson Inequality - I. Multiply each branch of
the well known Simpson kernel with an arbitrary quartic polynomial.

p(t) =


1

8!
(t− a)3(t− a+2b

3
)(t4 + a3t

3 + a2t
2 + a1t + a0), t ∈ [a, a+b

2
)

1

8!
(t− b)3(t− 2a+b

3
)(t4 + b3t

3 + b2t
2 + b1t + b0), t ∈ [a+b

2
, b]

Find the coefficients so that rule is corrected, that is end-point derivatives
are equal and opposite, and as many midpoint evaluations vanish as possi-
ble. The result is

(7.2)

∣∣∣∣∣
∫ b

a

f(t) dt−

[
b− a

294

(
59f(a) + 176f

(
a + b

2

)
+ 59f(b)

)
+

55

4166
(b− a)2

(
f ′(a)− f ′(b)

)
+

5

1029
(b− a)3f ′′

(
a + b

2

)
+

(b− a)4

30870

(
f ′′′(b)− f ′′′(a)

)]∣∣∣∣∣ ≤ 71

19914854400
(b− a)9‖f (8)‖∞

4. Eighth Order Modified Simpson Inequality - II. In an effort to elimi-
nate thef ′′((a + b)/2) term in (7.2), we replace the kernel above with the
following

p(t) =


1

8!
(t− a)4(t4 + a3t

3 + a2t
2 + a1t + a0), t ∈ [a, a+b

2
)

1

8!
(t− b)4(t4 + b3t

3 + b2t
2 + b1t + b0), t ∈ [a+b

2
, b]

This kernel eliminates sampling off (i), i = 8, 7, 6, 5 at t = a, b. Solve
for coefficients so that rule is corrected and as many midpoint evaluations

vanish as possible. The integration rule is

(7.3)

∣∣∣∣∣
∫ b

a

f(t) dt−

[
b− a

126

(
31f(a) + 64f

(
a + b

2

)
+ 31f(b)

)

+
5

252
(b− a)2

(
f ′(a)− f ′(b)

)
+

(b− a)4

15120

(
f ′′′(b)− f ′′′(a)

)]∣∣∣∣∣
≤ (b− a)9

101606400
‖f (8)‖∞

5. Order Ten Corrected Trapezoidal Rule Multiply the corrected trape-
zoidal kernel of Section 4 (see 3rd last equation in this section) with an
arbitrary sixth order polynomial.

p(t) =
1

10!
(t− a)2(t− b)2(t6 + a5t

5 + a4t
4 + a3t

3 + a2t
2 + a1t + a0)

Solve for coefficients so that rule is corrected, that is all end-point deriva-
tives are equal and opposite in sign. The resultant integral inequality is:

(7.4)

∣∣∣∣∣
∫ b

a

f(t) dt−

[
b− a

2

(
f(a) + f(b)

)
+

(b− a)2

12

(
f ′(a)− f ′(b)

)
+

(b− a)4

720

(
f ′′′(b)− f ′′′(a)

)
+

(b− a)6

30240

(
f (5)(a)− f (5)(b)

)
+

(b− a)8

1209600

(
f (7)(b)− f (7)(a)

)]∣∣∣∣∣ ≤ (b− a)11

47900160
‖f (10)‖∞

[14] J. PĚCARIĆ AND S. VAROS̆ANEC, A note on Simpson’s inequality for
functions of bounded variation,Tamkang J. Math., 31(3) (2000) 239–242.

http://jipam.vu.edu.au/
mailto:John.Roumeliotis@vu.edu.au
http://jipam.vu.edu.au/


Integral Inequalities and
Computer Algebra Systems

John Roumeliotis

Title Page

Contents

JJ II

J I

Go Back

Close

Quit

Page 44 of 44

J. Ineq. Pure and Appl. Math. 6(5) Art. 141, 2005

http://jipam.vu.edu.au

